* Figure: CONTINUED ADOPTION AND USE OF INTERVENTION/IMPROVED STOVES

use "${output}panel_r0_r1_r2.dta", clear // Load merged analysis panel

*********************

label define ngo_status 0 "Non-NGO" 1 "NGO"
label values chirag_strata ngo_status
label define surveyround_status 0 "Baseline" 1 "Midline" 2 "Endline"
label value surveyround surveyround_status

*********************

* Intervention stove ownership (%)
clonevar intervention_stove_own_pc = intervention_stove_own
replace intervention_stove_own_pc = intervention_stove_own_pc * 100
cibar intervention_stove_own_pc, over1(chirag_strata) over2(surveyround) ///
	bargap(10) ciopts(msize(0) lcolor(black)) ///
	baropts(base(0) yla(0(10)50)) ///
	graphopts(title("(a) Own an intervention stove") ///
		ytitle("") legend(col(2)) name(g_intervention_own, replace) nodraw)
	
* Intervention stove use (%)
clonevar intervention_stove_use_pc = intervention_stove_use
replace intervention_stove_use_pc = intervention_stove_use_pc * 100
cibar intervention_stove_use_pc, over1(chirag_strata) over2(surveyround) ///
	bargap(10) ciopts(msize(0) lcolor(black)) ///
	baropts(base(0) yla(0(10)50)) ///
	graphopts(title("(b) Use an intervention stove") ///
		ytitle("") name(g_intervention_use, replace) nodraw)

* Improved stove ownership (%)
clonevar improved_stove_own_pc = improved_stove_own
replace improved_stove_own_pc = improved_stove_own_pc * 100
cibar improved_stove_own_pc, over1(chirag_strata) over2(surveyround) ///
	bargap(10) ciopts(msize(0) lcolor(black)) ///
	baropts(base(0) yla(0(10)70)) ///
	graphopts(title("(c) Own an improved stove") ///
		ytitle("") name(g_improved_own, replace) nodraw)
		
* Improved stove use (%)
clonevar improved_stove_use_pc = improved_stove_use
replace improved_stove_use_pc = improved_stove_use_pc * 100
cibar improved_stove_use_pc, over1(chirag_strata) over2(surveyround) ///
	bargap(10) ciopts(msize(0) lcolor(black)) ///
	baropts(base(0) yla(0(10)70)) ///
	graphopts(title("(d) Use an improved stove") ///
		ytitle("") name(g_improved_use, replace) nodraw) 
		
grc1leg g_intervention_own g_intervention_use g_improved_own g_improved_use, ///
	legendfrom(g_intervention_own) name(g_combined, replace) iscale(*.8) l1("Percent of households")

graph display g_combined, xsize(12)  ysize(15)

graph export "${results}figure_continued_adoption_use.png", width(5000) replace
graph drop _all
